

Le grand livre de eForth web

version 1.1 - 29 octobre 2023



Auteur(s)

- Marc PETREMANN

Collaborateur(s)

- xxx

Table des matières

Premiers pas avec eFORTH web.....	3
Premiers mots et premières définitions.....	3
Voir les autres vocabulaires.....	3
Première compilation de code FORTH.....	4
Sélection mode texte ou graphique.....	4
Dictionnaire / Pile / Variables / Constantes.....	6
Étendre le dictionnaire.....	6
Gestion du dictionnaire.....	6
Piles et notation polonaise inversée.....	7
Manipulation de la pile de paramètres.....	8
La pile de retour et ses utilisations.....	8
Utilisation de la mémoire.....	9
Variables.....	9
Constantes.....	10
Valeurs pseudo-constantes.....	10
Outils de base pour l'allocation de mémoire.....	10
Les nombres entiers avec eFORTH.....	12
Empilage et dépilage des nombres entiers.....	12
Opérations arithmétiques élémentaires.....	13
Somme de deux nombres entiers.....	13
Soustraction de deux nombres entiers.....	14
Produit de deux nombres entiers.....	14
Quotient de deux nombres entiers.....	14
Produit et quotient de trois nombres.....	15
Traitement des expressions algébriques.....	15
Manipulation des données sur la pile.....	16
Passage de paramètres par la pile de retour.....	17
Import de fichiers vers eFORTH web.....	19
Edition des fichiers sources pour eFORTH.....	19
Chargement d'un fichier source vers eForth.....	19
Créez des définitions FORTH en utilisant JavaScript.....	21
Le mariage Javascript Forth.....	21
Structure d'une définition de mot avec JSWORD:.....	21
Passage des paramètres entre eFORTH et JavaScript.....	22
Extension du vocabulaire web de eFORTH.....	24

Premiers pas avec eFORTH web

Premiers mots et premières définitions

Si vous n'avez pas installé eFORTH web, vous pouvez tester notre version en ligne. Lien :

<https://eforth.arduino-forth.com/web-eforth/index/>

Pour constater le bon fonctionnement de eFORTH, tapez **words** dans l'espace de travail eFORTH:

Essayez eFORTH avec la version en ligne

```
uEforth v7.0.7.15 - rev 564a8fc68b545ebeb3ab
Forth dictionary: 4065196 free + 79456 used = 4144652 total (98% free)
3 x Forth stacks: 16384 bytes each
ok
--> words
FORTH ok colors ms start-task task pause tasks END OF ENDCASE CASE +to
to exit ; { (local) asm words vlist order see .s startswith? str= :noname
forget dump spaces assert set-title page at-xy normal bg fg ansi ms-ticks
web structures f.s f. #fs set-precision precision fvariable fconstant fliteral
afliteral sf, internals sealed previous also only transfer{ }transfer transfer
definitions vocabulary [IF] [ELSE] [THEN] DEFINED? quit evaluate prompt
refill tib accept echo abort" abort z>s s>z r~ r| r" z" ." s" n. ? . u.
binary decimal octal hex str #> sign #s # hold <# extract pad hld cr space
emit bye terminate key? key type fill32 >name is defer +to to value throw
catch handler K J I loop +loop leave UNLOOP ?do do next for nest-depth
postpone recurse aft repeat while else if then ahead until again begin
[char] char [']' ' used remaining fdepth depth fp0 rp0 sp0 >link >link&
>size >params >name-length >flags >flags& align aligned #! \ ( CALL FP@
FP! SF@ SF! FDUP FNIP FDROP FOVER FSWAP FROT FNEGATE F0< F0= F= F< F> F<>
```

L'exécution de **words** affiche le contenu du vocabulaire FORTH. Tous ces mots sont l'équivalent de fonctions dans d'autres langages. FORTH se différencie de tous les autres langages de programmation, car en FORTH toute nouvelle définition de mot FORTH agrandit le langage FORTH. Il n'y a pas de différence entre une application en FORTH et l'extension du dictionnaire.

Pour effacer le contenu de l'espace de travail, exécutez le mot **page**.

Voir les autres vocabulaires

words affiche le contenu du vocabulaire FORTH. Mais eFORTH web embarque d'autres vocabulaires. Pour voir la liste de ces vocabulaires, tapez **internals voclist**. Affiche:

```
--> internals voclist
tasks
ansi
web
structures
internalized
internals
FORTH
```

```
ok
-->
```

Parmi ces vocabulaires, celui qui va nous intéresser est le vocabulaire **web**. Pour voir les définitions de ce vocabulaire, taper **web vlist**. Affiche:

```
--> web vlist
yielding-task yielding import rm ls download cat include-file upload upload-file
session? web-key? web-key web-type scripts scripts# random button mouse
textWidth fillText font text-size! log raw-http-upload http-download raw-download
upload-success? upload-done? upload-start ms-ticks silence tone importScripts
release keyCount getKey clearItems removeItem getItem setItem smooth gpop
gpush rotate scale translate show-text keys-height mobile textRatios viewport@
window line fill stroke lineTo moveTo beginPath box lineWidth color! text
gr grmode shouldEcho? web-terminate web-key?-raw web-key-raw web-type-raw
jseval JSWORD: jsslot jseval!
ok
```

Un vocabulaire permet d'intégrer des mots utilisables dans un certain contexte. Ici, le contenu du vocabulaire web n'est défini que pour la version eFORTH web. Le vocabulaire web n'existe pas sur les autres versions de eFORTH (Windows, Linux, ESP32...).

Première compilation de code FORTH

Si une définition FORTH est courte, vous pouvez la compiler immédiatement depuis l'invite de commande de eFORTH web:

```
: myLoop 10 0 do i . loop ;
```

Tapez ensuite **myLoop**, ce qui exécutera le code qui vient d'être compilé par eFORTH web:

```
uEforth v7.0.7.9 - rev fb3db70da6d111b1fdf0
Forth dictionary: 4068200 free + 76708 used = 4144908 total (98% free)
3 x Forth stacks: 16384 bytes each
ok
--> : myLoop 10 0 do i . loop ;
ok
--> myloop
0 1 2 3 4 5 6 7 8 9 ok
-->
```

Sélection mode texte ou graphique

Le mot **gr** sélectionne le mode graphique. Ce mot rend visible un espace graphique de type canvas:

```
web gr
```

Pour revenir au mode text, taper **text**.


Voici un exemple de graphisme:

```
web
: grTest ( -- )
  gr 400 300 window
```

```
$000000 color!  
5 295 395 295 line  
5 295 5 5 line  
$ff0000 color! 10 294 10 -30 box  
$00ff00 color! 30 294 10 -80 box  
$ffff00 color! 50 294 10 -120 box  
$0000ff color! 70 294 10 -45 box  
key drop  
text  
;
```

L'exécution de **grTest** affiche ce graphisme:

Try eFORTH



uForth v7.0.7.9 - rev fh3dh70da6d111b1fdf0

Dictionnaire / Pile / Variables / Constantes

Étendre le dictionnaire

Forth appartient à la classe des langages d'interprétation tissés. Cela signifie qu'il peut interpréter les commandes tapées sur la console, ainsi que compiler de nouveaux sous-programmes et programmes.

Le compilateur Forth fait partie du langage et des mots spéciaux sont utilisés pour créer de nouvelles entrées de dictionnaire (c'est-à-dire des mots). Les plus importants sont **:** (commencer une nouvelle définition) et **;** (termine la définition). Essayons ceci en tapant:

```
: *+ * + ;
```

Ce qui s'est passé? L'action de **:** est de créer une nouvelle entrée de dictionnaire nommée ***+** et passer du mode interprétation au mode compilation. En mode compilation, l'interpréteur recherche les mots **et**, plutôt que de les exécuter, installe des pointeurs vers leur code. Si le texte est un nombre, au lieu de le pousser sur la pile, ESP32forth construit le nombre dans le dictionnaire l'espace alloué pour le nouveau mot, suivant le code spécial qui met le numéro stocké sur la pile chaque fois que le mot est exécuté. L'action d'exécution de ***+** est donc d'exécuter séquentiellement les mots définis précédemment ***** et **+**.

Le mot **;** est spécial. C'est un mot immédiat et il est toujours exécuté, même si le système est en mode compilation. Ce que fait **;** est double. Tout d'abord, il installe le code qui renvoie le contrôle au niveau externe suivant de l'interpréteur et, deuxièmement, il revient du mode compilation au mode interprétation.

Maintenant, essayez votre nouveau mot :

```
decimal 5 6 7 *+ . \ affiche 47 ok<#,ram>
```

Cet exemple illustre deux activités principales de travail dans Forth: ajouter un nouveau mot au dictionnaire, et l'essayer dès qu'il a été défini.

Gestion du dictionnaire

Le mot **forget** suivi du mot à supprimer enlèvera toutes les entrées de dictionnaire que vous avez faites depuis ce mot:

```
: test1 ;  
: test2 ;  
: test3 ;  
forget test2 \ efface test2 et test3 du dictionnaire
```


Notez que ESP32forth affiche commodément les éléments de la pile lors de l'interprétation de chaque ligne et que la valeur de -16 est affichée sous la forme d'entier non signé 32 bits. En outre, le mot `.` consomme la valeur de données -104, laissant la pile vide. Si nous exécutons `.` sur la pile maintenant vide, l'interpréteur externe abandonne avec une erreur de pointeur de pile `STACK UNDERFLOW ERROR`.

La notation de programmation où les opérandes apparaissent en premier, suivis du ou des opérateurs est appelée Notation polonaise inverse (RPN).

Manipulation de la pile de paramètres

Étant un système basé sur la pile, ESP32forth doit fournir des moyens de mettre des nombres sur la pile, pour les supprimer et réorganiser leur ordre. On a déjà vu qu'on peut mettre des nombres sur la pile simplement en les tapant. Nous pouvons également intégrer les nombres dans la définition d'un mot FORTH.

Le mot **drop** supprime un numéro du sommet de la pile mettant ainsi le suivant au sommet. Le mot **swap** échange les 2 premiers numéros. **dup** copie le nombre au sommet, poussant tout les autres numéros vers le bas. **rot** fait pivoter les 3 premiers nombres. Ces



actions sont présentées ci-dessous.

La pile de retour et ses utilisations

Lors de la compilation d'un nouveau mot, ESP32forth établit des liens entre le mot appelant et les mots définis précédemment qui doivent être invoqués par l'exécution du nouveau mot. Ce mécanisme de liaison, lors de l'exécution, utilise la pile de retour (rstack). L'adresse du mot suivant à invoquer est placée sur la pile de retour de sorte que, lorsque le mot courant est terminé en cours d'exécution, le système sait où passer au mot suivant. Comme les mots peuvent être imbriqués, il doit y avoir une pile de ces adresses de retour.

En plus de servir de réservoir d'adresses de retour, l'utilisateur peut également stocker et récupérer à partir de la pile de retour, mais cela doit être fait avec soin car la pile de retour est essentielle à l'exécution du programme. Si vous utilisez la pile de retour pour le

stockage temporaire, vous devez la remettre dans son état d'origine, sinon vous ferez probablement planter le système ESP32forth. Malgré le danger, il y a des moments où l'utilisation de pile de retour comme stockage temporaire peut rendre votre code moins complexe.

Pour stocker dans la pile, utilisez **>r** pour déplacer le sommet de la pile de paramètres vers le haut de la pile de retour. Pour récupérer une valeur, **r>** déplace la valeur supérieure de la pile de retour vers le sommet de la pile de paramètres. Pour supprimer simplement une valeur du haut de la pile, il y a le mot **rdrop**. Le mot **r@** copie le haut de la pile de retour dans la pile de paramètres.

Utilisation de la mémoire

Dans eForth WEB, les nombres 32 bits sont extraits de la mémoire vers la pile par le mot **@** (fetch) et stocké du sommet à la mémoire par le mot **!** (store). **@** attend une adresse sur la pile et remplace l'adresse par son contenu. **!** attend un nombre et une adresse pour le stocker. Il place le numéro dans l'emplacement de mémoire référencé par l'adresse, consommant les deux paramètres dans le processus.

Les nombres non signés qui représentent des valeurs de 8 bits (octets) peuvent être placés dans des caractères de la taille d'un caractère. cellules de mémoire en utilisant **c@** et **c!**.

```
create testVar
  cell allot
  $f7 testVar c!
testVar c@ . \ affiche 247
```

Variables

Une variable est un emplacement nommé en mémoire qui peut stocker un nombre, tel que le résultat intermédiaire d'un calcul, hors de la pile. Par exemple:

```
variable x
```

crée un emplacement de stockage nommé, **x**, qui s'exécute en laissant l'adresse de son emplacement de stockage au sommet de la pile:

```
x . \ affiche l'adresse
```

Nous pouvons alors aller chercher ou stocker à cette adresse :

```
variable x
3 x !
x @ . \ affiche: 3
```

Constantes

Une constante est un nombre que vous ne voudriez pas changer pendant l'exécution d'un programme. Le résultat de l'exécution du mot associé à une constante est la valeur des données restant sur la pile.

```
\ définit les pins VSPI
19 constant VSPI_MISO
23 constant VSPI_MOSI
18 constant VSPI_SCLK
05 constant VSPI_CS

\ définit la fréquence du port SPI
4000000 constant SPI_FREQ

\ sélectionne le vocabulaire SPI
only FORTH SPI also

\ initialise le port SPI
: init.VSPI ( -- )
  VSPI_CS OUTPUT pinMode
  VSPI_SCLK VSPI_MISO VSPI_MOSI VSPI_CS SPI.begin
  SPI_FREQ SPI.setFrequency
;
```

Valeurs pseudo-constantes

Une valeur définie avec **value** est un type hybride de variable et constante. Nous définissons et initialisons une valeur et est invoquée comme nous le ferions pour une constante. On peut aussi changer une valeur comme on peut changer une variable.

```
decimal
13 value thirteen
thirteen . \ display: 13
47 to thirteen
thirteen . \ display: 47
```

Le mot **to** fonctionne également dans les définitions de mots, en remplaçant la valeur qui le suit par tout ce qui est actuellement au sommet de la pile. Vous devez faire attention à ce que **to** soit suivi d'une valeur définie par **value** et non d'autre chose.

Outils de base pour l'allocation de mémoire

Les mots **create** et **allot** sont les outils de base pour réserver un espace mémoire et y attacher une étiquette. Par exemple, la transcription suivante montre une nouvelle entrée de dictionnaire **graphic-array** :

```
create graphic-array ( --- addr )
  %00000000 c,
```

```
%00000010 c,  
%00000100 c,  
%00001000 c,  
%00010000 c,  
%00100000 c,  
%01000000 c,  
%10000000 c,
```

Lorsqu'il est exécuté, le mot **graphic-array** poussera l'adresse de la première entrée.

Nous pouvons maintenant accéder à la mémoire allouée à **graphic-array** en utilisant les mots de récupération et de stockage expliqués plus tôt. Pour calculer l'adresse du troisième octet attribué à **graphic-array** on peut écrire **graphic-array 2 +**, en se rappelant que les indices commencent à 0.

```
30 graphic-array 2 + c!  
graphic-array 2 + c@ . \ affiche 30
```

Les nombres entiers avec eFORTH

Empilage et dépilage des nombres entiers

Le langage FORTH stocke les nombres sur une pile nommée pile de données ou pile paramétrique. L'empilage d'un nombre est très simple:

```
55
```

empile le nombre 55. Pour dépiler ce nombre, il y a plusieurs méthodes. La plus simple consiste à afficher le contenu du dernier élément empilé. Le mot `.` (point) affiche 55.

```
55      \ display ok
.       \ display 55 ok
```

Si vous déposez plusieurs nombres sur la pile de données, voici ce qui se passe:

```
11 45 6543
```

Les nombres sont mis sur la pile de données dans l'ordre de frappe. Le dernier nombre entré est toujours au sommet de la pile:

valeur empilée	haut -- pile -- fond
11	11
45	45 11
6543	6543 45 11

Le dépilage successif des nombres affiche ceux-ci dans l'ordre inverse de leur empilage:

```
11 45 6543  \ display ok
.           \ display 6543 ok
.           \ display 45  ok
.           \ display 11  ok
```

Pour mieux visualiser le mécanisme d'empilage et de dépilage des nombres, pensez à une pile d'assiettes: la dernière assiette déposée sur la pile sera la première reprise.

A tout moment vous pouvez prendre connaissance du contenu de la pile de données sans avoir à provoquer le dépilage des valeurs qui y sont stockées en utilisant le mot `.S`:

```
1 2 3 .S affiche 1 2 3
```

Ce principe de rangement est appelé également pile LIFO (Last In, First Out) dans certains ouvrages écrits en anglais pour désigner une pile dont le mécanisme est: "dernier entré, premier sorti".

Avec la majorité des versions du langage FORTH, la quantité de nombres empilables est assez élevée, mais reste limitée. Si vous empilez trop de nombres, vous saturerez la pile de données. De même, toute tentative pour dépiler un nombre alors que la pile de données est vide, affichera un message d'erreur.

Chaque nombre empilé ne peut être qu'un nombre entier au format simple précision. Selon les cas, ce nombre peut être considéré signé ou non signé:

- signé ou non signé: paramètre de calcul arithmétique
- non signé: adresse simple précision

En format simple précision signé, le bit de poids le plus fort indique le signe du nombre.

La valeur -1, déposée sur la pile en tant que valeur entière a la représentation binaire suivante:

```
11111111111111111111111111111111 (simple précision 32 bits) .
```

Cette même valeur peut être affichée en valeur absolue en utilisant le mot **U.** à la place du mot **.** (point):

```
-1 U. \ affiche 4294967295
```

Opérations arithmétiques élémentaires

Les opérateurs arithmétiques **+** **-** ***** et **/** agissent sur les deux valeurs situées au sommet de la pile de données. Les valeurs traitées sont toujours des entiers simple précision signés.

Somme de deux nombres entiers

Pour additionner deux nombres, il faut d'abord les déposer sur la pile de données:

```
22 44 + . \ affiche 66
```

Une fois empilés 22 et 44, le mot **+** opère l'addition de ces deux valeurs et le mot **.** affiche le résultat:

```
55 1 + 3 + . \ affiche 59 et peut aussi s'écrire  
55 1 3 + + .
```

L'addition est commutative: les valeurs peuvent être déposées sur la pile de données dans n'importe quel ordre:

```
55 22 +  
22 55 +
```

Ce principe de calcul est appelé NOTATION POLONAISE INVERSE (RPN dans la littérature anglaise, pour Reverse Polish Notation). On peut aussi additionner deux nombres entiers non signés, à condition de visualiser le résultat à l'aide du mot **U.** au lieu de **.**:

```
35000 10 + U.
```

Il est également possible de faire la somme de deux nombres de signe différent:

```
10 -5 + . \ affiche 5  
-5 10 + . \ affiche aussi 5
```

FORTH dispose également du mot **1+** qui incrémente la valeur située au sommet de la pile de données de 1 unité:

```
10 1+      \ est équivalent à
10 1 +
```

Soustraction de deux nombres entiers

Soit deux nombres a et b. La différence de deux nombres sera écrite en FORTH sous la forme:

```
a b - pour a-b
```

La soustraction n'est pas commutative:

```
10 3 - .      \ affiche      7
3 10 - .      \ affiche     -7
```

Le mot **1-** décrémente la valeur située au sommet de la pile de données de 1 unité:

```
10 1-      \ est équivalent à
10 1 -
```

Produit de deux nombres entiers

Soit deux nombres a et b. Le produit de deux nombres sera écrit en FORTH sous la forme:

```
a b *      \ pour a*b
```

La multiplication est commutative:

```
7 5 * .      \ ou
5 7 * .      \ affiche    35
```

Le mot **2*** multiplie la valeur située au sommet de la pile de données par deux:

```
5 2*      \ est équivalent à
5 2 *
```

Quotient de deux nombres entiers

Pour la division, seul le quotient entier est conservé sur la pile de données:

```
22 7 / .      \ affiche    3
```

La division n'est pas commutative:

```
15 5 / .      \ affiche    3
5 15 / .      \ affiche    0
```

Le reste de la division peut être obtenu en appliquant la fonction modulo:

```
22 7 MOD .      \ affiche    1
```

La fonction modulo peut servir à déterminer la divisibilité d'un nombre par un autre:

```

: DIV? ( n1 n2 ---)
  OVER OVER MOD CR
  IF
    SWAP . ." n'est pas "
  ELSE
    SWAP . ." est "
  THEN
    ." divisible par " . ;

```

Le mot **/MOD** combine les actions de **/** et de **MOD**:

```

22 7 /MOD . . \ affiche 3 1

```

Produit et quotient de trois nombres

Le mot ***/** combine les opérations de multiplication et de division.

Exemple, soit à calculer le prix TTC d'une marchandise (TVA à 20 %), on définira le mot TTC comme suit:

```

: TTC ( n1 --- n2)
  DUP 200 1000 */ + ;
442 TTC . \ affiche 530

```

Les valeurs traitées étant exprimées en centimes.

Le mot ***/MOD** a les mêmes propriétés que ***/**, mais délivre le quotient et le reste de l'opération.

A titre d'exemple, pour donner une application immédiate et pratique des notions déjà exprimées, est la conversion des degrés Fahrenheit et Celsius:

- la conversion des degrés Fahrenheit en degrés Celsius obéit à la formule
 $^{\circ}\text{C} = (^{\circ}\text{F} - 32) * 5/9$
- la conversion des degrés Celsius en degrés Fahrenheit obéit à la formule
 $^{\circ}\text{F} = 9/5 * ^{\circ}\text{C} + 32$

```

: C>F ( °C --- °F)
  9 5 */ 32 + ;
: F>C ( °F --- °C)
  32 - 5 9 */ ;
37 C>F . \ affiche 98 (les résultats sont arrondis)

```

Cet exemple fonctionne sur toutes les versions du langage FORTH.

Traitement des expressions algébriques

Les opérations peuvent être chaînées, mais une opération en notation algébrique comportant des parenthèses doit être convertie en notation RPN en tenant compte de

l'ordre de priorité des opérations. FORTH n'utilise pas les parenthèses dans les opérations arithmétiques:

- soit l'expression algébrique $(2 + 5) * (7 - 2)$
- elle s'écrit en FORTH `2 5 + 7 2 - *`

Lors d'une opération de conversion de notation algébrique infixée en notation polonaise inverse, commencez toujours par le niveau de parenthèse le plus imbriqué et par la gauche. Ecrivez la transcription en notation polonaise inverse de chaque opération sur des lignes séparées, successivement de haut en bas, en les mettant dans le prolongement de l'expression algébrique exprimée dans la formule initiale.

C'est choquant? Mais tous les interpréteurs/compilateurs travaillent ainsi lorsqu'ils ont à évaluer une formule algébrique. En notation algébrique, les parenthèses ne servent qu'à isoler une expression sous forme de sous-expression qui devient membre d'une expression plus générale.

En informatique comme en arithmétique, un opérateur travaille toujours sur deux opérandes et seulement deux opérandes simultanément. Le résultat d'une opération portant sur deux opérandes délivre une valeur qui peut devenir à son tour opérande d'un autre opérateur. L'ordre d'exécution des opérandes et des opérateurs est fondamental:

notation algébrique	polonaise inverse
$(2+3)*5$	2 3 + 5 *
$2+(3*5)$	2 3 5 * + or 3 5 * 2 +

Tous les problèmes arithmétiques peuvent être résolus de cette manière, ce n'est qu'une question d'habitude. L'exemple donné précédemment illustre parfaitement la rigueur dont doit faire preuve le programmeurs Forth. Cette rigueur garantit un fonctionnement sans ambiguïté des programmes, quel que soit leur niveau de complexité.

Manipulation des données sur la pile

La pile de données est l'élément fondamental du langage FORTH pour le traitement de données. Son fonctionnement est identique à celui de la pile gérée par le micro-processeur. Dans certaines situations, les données traitées par les différentes définitions doivent être réordonnées ou dupliquées.

Le mot **DUP** duplique le contenu du sommet de la pile de données:

```
10 DUP . . \ affiche 10 10
5 DUP * . \ affiche 25
```

Le mot **OVER** duplique le second élément de la pile de données:

```
5 15 OVER . . . \ affiche 5 15 5
```

Le mot **SWAP** inverse les deux éléments du sommet de la pile de données:

```
1 3 SWAP . . \ affiche 1 3
```


Le mot **ROT** effectue une rotation sur les trois éléments situés au sommet de la pile de données:

```
1 2 3 ROT . . . \ affiche 1 3 2
```

Voici quelques exemples d'utilisation de ces manipulateurs de pile de données:

```
: AU-CARRE ( n --- n2)
  DUP * ;
```

Le mot **AU-CARRE** élève un nombre entier quelconque au carré:

```
2 AU-CARRE . \ affiche 4
3 AU-CARRE . \ affiche 9
4 AU-CARRE . \ affiche 16

: AU-CUBE ( n --- n3)
  DUP DUP * * ;
```

Le mot **AU-CUBE** élève un nombre entier quelconque au cube:

```
2 AU-CUBE . \ affiche 8
3 AU-CUBE . \ affiche 27
```

Attention, n'utilisez pas des valeurs trop élevées, car un résultat dépassant 2 puissance 32, pour une pile de données 32 bits, devient faux. Ayez toujours à l'esprit que les données traitées sont des entiers, donc de capacité limitée.

Passage de paramètres par la pile de retour

A côté de la pile de paramètres, il existe dans FORTH une deuxième pile, appelée pile de retour parce qu'elle sert à l'interpréteur interne à retrouver l'adresse de retour à chaque appel d'une procédure.

Il y a parfois des cas extrêmes où on peut être amené à stocker un ou plusieurs paramètres ailleurs que sur la pile de données, ceci pour simplifier quelque peu certaines manipulations scabreuses. la solution la plus commode, parmi d'autres, est la pile de retour. Cette pile est accessible par les mots **>R** et **R>** moyennant quelques précautions pour ne pas compromettre le fonctionnement de cette pile interne.

Le mot **>R** transfère un nombre entier de la pile de données vers la pile de retour.

Le mot **R>** transfère un nombre entier de la pile de retour vers la pile de données.

Une opération **>R R>** est nulle. En fin de définition, il doit y avoir autant de **>R** que de **R>** sous peine de perturber quelque peu le déroulement normal de votre définition. Exemple d'utilisation:

```
: AU-CARRE ( n --- n^2)          \ élévation au carré
  DUP >R                        \ transfert de la copie sur pile de
retour                           retour
  CR ." Le carré de " .         \ affichage valeur initiale
```

<code>."</code> est " <code>R> DUP * . ;</code>	<code>\ récupération valeur déposée sur pile</code>
<code>retour</code>	<code>\ et affichage de son carré</code>

Import de fichiers vers eFORTH web


Vous avez installé eFORTH sur votre site web? Ou en local au pire des cas?

OK.

Voyons donc les choses sérieuses. FORTH est intéressant si on développe de vrais programmes.

Edition des fichiers sources pour eFORTH

Les fichiers sources pour eFORTH peuvent être édités avec n'importe quel éditeur texte: Wordpad, bloc-notes...



```
\ locals variables in DUMP:
\ START_ADDR      \ first address to dump
\ END_ADDR        \ latest address to dump
\ ØSTART_ADDR     \ first address for dump loop
\ LINES           \ number of lines for dump loop
\ myBASE          \ current numeric base

: dump ( start len -- )
  cr cr ." --addr--- "
  ." 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F  -----chars-----"
  2dup + { END_ADDR }          \ store latest address to dump
  swap { START_ADDR }         \ store START address to dump
  START_ADDR 16 / 16 * { ØSTART_ADDR } \ calc. addr for loop start
  16 / 1+ { LINES }
  base @ { myBASE }           \ save current base
  hex
  \ outer loop
  LINES 0 do
    ØSTART_ADDR + 16 * + \ calc. start address for current line
```

Ici, édition du fichier **dumpTool.fs** avec le bloc-notes.


Sauvegardez vos fichiers source dans le répertoire de votre choix.

Chargement d'un fichier source vers eForth

Pour charger le contenu de notre fichier **dumpTool.fs**, il suffit d'exécuter ces deux mots depuis l'invite de commande de eFORTH:

```
web import
```

La fenêtre de sélection de fichier s'ouvre sur votre ordinateur:



Ici, on sélectionne le fichier **dumpTool.fs** précédemment édité et sauvegardé.

L'envoi du fichier provoque immédiatement le traitement de son contenu. Ici, on a compilé le mot **dump** qu'il suffit ensuite de tester:

```
ok
--> here 100 dump
--addr--- 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F -----chars-----
0001-F450 D4 90 01 00 00 00 00 00 00 00 00 00 00 00 00 .....
0001-F460 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 .....
0001-F470 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 .....
0001-F480 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 .....
0001-F490 00 00 00 00 00 00 00 00 00 00 00 30 30 30 31 2D .....0001-
0001-F4A0 46 34 33 33 00 00 00 00 00 00 00 00 00 00 00 F400.....
0001-F4B0 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 .....
ok
-->
```

Un fichier source pour eFORTH peut contenir des mots à compiler, mais aussi des commandes, dont l'exécution des mots précédemment compilés. Cette facilité permet de gagner un temps considérable en temps de développement.

Seule différence, avec eFORTH pour windows, les fichiers source ne peuvent pas contenir un appel de fichier par include.

Créez des définitions FORTH en utilisant JavaScript

Le mariage Javascript Forth

eFORTH web fonctionne dans tous les navigateurs web, sur tous les systèmes Windows, Linux, MacOS, Android... Comment est-ce possible?

Le secret de cette extraordinaire polyvalence est très simple: un mariage réussi entre Javascript et FORTH.

La plupart des primitives du langage FORTH sont écrites dans un pseudo-assembleur ou en FORTH. Nous allons voir comment agrandir le vocabulaire web en définissant le mot `ellipse`:

```
web definitions

\ draw ellipse in graphic mode
JSWORD: ellipse { x y rx ry angle div }
  context.ctx.ellipse(x, y, rx, ry, Math.PI * 2 * angle / div, 0, 2 *
Math.PI);
~
forth definitions
```

Ici, le mot `ellipse` est défini après le mot de création `JSWORD:`. Le délimiteur de fin de définition est marqué par le caractère `~`. Voyons en détail la structure de définition de notre mot `ellipse`.

Structure d'une définition de mot avec JSWORD:

Commençons par analyser la première ligne de cette définition de `ellipse`:

```
JSWORD: ellipse { x y rx ry angle div }
  context.ctx.ellipse(x, y, rx, ry, Math.PI * 2 * angle / div, 0, 2 *
Math.PI);
~
```

- le mot `JSWORD:` marque le début d'une définition utilisant le code Javascript. Ce mot de définition est immédiatement suivi du mot à définir, ici `ellipse`;
- le mot `ellipse` est le mot en cours de définition. Il est suivi d'un passage de paramètres par variables locales;
- les paramètres sont passés par `{ x y rx ry angle div }`.

La ligne suivante contient le code Javascript:

```
JSWORD: ellipse { x y rx ry angle div }
  context.ctx.ellipse(x, y, rx, ry, Math.PI * 2 * angle / div, 0, 2 *
Math.PI);
~
```

Le code Javascript peut être écrit sur plusieurs lignes:

```
web definitions
JSWORD: time@ { -- h m s }
  let date = new Date();
  var hh = date.getHours();
  var mm = date.getMinutes();
  var ss = date.getSeconds();
  return [hh, mm, ss];
~

JSWORD: date@ { -- y m d }
  let date = new Date();
  var yy = date.getFullYear();
  var mm = date.getMonth()+1;
  var dd = date.getUTCDate();
  return [yy, mm, dd];
~
forth definitions
```

Nous allons voir en détail comment s'effectue le passage des paramètres.


Passage des paramètres entre eFORTH et JavaScript

Le passage de paramètres exploite les variables locales. Ces variables sont définies entre le mot **{** et le caractère **}**. Reprenons l'exemple de la définition du mot ellipse:

```
JSWORD: ellipse { x y rx ry angle div }
  context.ctx.ellipse(x, y, rx, ry, Math.PI * 2 * angle / div, 0, 2 *
Math.PI);
~
```

Ici nous passons cinq paramètres. Ces paramètres sont ceux utilisés ensuite dans une définition en langage FORTH:

```
web gr
600 300 window
$ff0000 color!
100 100 75 30 0 360 ellipse fill
0 color! stroke
```



Il y a quatre cas de passage de paramètres à un mot défini par **JSWORD** :

1) Appel à une fonction JavaScript sans passage de paramètre. Les accolades sont vides:

```
JSWORD: beginPath { }  
    context.ctx.beginPath();  
~
```

2) Appel à une fonction JavaScript avec passage de paramètres en entrée seulement:

```
JSWORD: lineTo { x y }  
    context.ctx.lineTo(x, y);  
~
```

3) Appel à une fonction JavaScript avec récupération de paramètres en sortie seulement:

```
JSWORD: viewport@ { -- w h }  
    return [context.width, context.height];  
~
```

4) Et enfin, passage de paramètres en entrée et en sortie:

```
JSWORD: web-type-raw { a n -- yld }  
    if (globalObj.write) {  
        var text = GetString(a, n);  
        write(text);  
        return 0;  
    } else {  
        var newline = false;  
        for (var i = 0; i < n; ++i) {  
            var ch = u8[a + i];  
            if (ch == 10) { newline = true; }  
            context.Emit(ch);  
        }  
        if (newline) {  
            context.Update();  
        }  
        return newline ? -1 : 0;  
    }
```

```
}  
~
```

Dans ce dernier cas, on sépare les paramètres d'entrée et de sortie avec `--`. En temps normal, dans une définition eFORTH classique, ces caractères et ceux qui suivent sont ignorés. Pour **JSWORD**, le passage des paramètres doit impérativement respecter une des quatre formes énumérées ci-avant.

Pour les paramètres de sortie, il n'est pas nécessaire d'utiliser des variables JavaScript. Reprenons le code de **time@**:

```
web definitions  
JSWORD: time@ { -- h m s }  
  let date = new Date();  
  var hh = date.getHours();  
  var mm = date.getMinutes();  
  var ss = date.getSeconds();  
  return [hh, mm, ss];  
~  
forth definitions
```

Ce code peut être réécrit comme suit:

```
web definitions  
JSWORD: time@ { -- h m s }  
  let date = new Date();  
  return [date.getHours(), date.getMinutes(), date.getSeconds()];  
~  
forth definitions
```

La fonction JavaScript **return()** doit simplement énumérer les valeurs à passer à eFORTH. Les `--` doivent être suivis d'autant de paramètres que ceux renvoyés par **return()**.

Il est facile d'étendre eFORTH en rajoutant des fonctions de tout type: file, HTTP, etc...

Extension du vocabulaire web de eFORTH

Le vocabulaire web est très incomplet. Vous pouvez l'étendre en récupérant le contenu du fichier **additionalDefs.fs** dans le fichier **eForthWEB-book.zip**.

Installez ce fichier dans le répertoire de votre choix. Pour notre part, nous avons une préférence pour le nom de répertoire **fs** (pour Forth Script). Pour appeler le contenu de ce fichier **additionalDefs.fs** depuis votre application web:

```
<div id="ueforth"></div>  
<script src="../js/ueforth.js"></script>  
<script type="text/forth" src="fs/additionnalDefs.fs"></script>  
<script type="text/forth">  
  \ here additionnal Forth Code
```



```
</script>
```

Index lexical

1-.....14	mémoire.....9	:.....6
1+.....14	MOD.....14	.S.....12
2*.....14	OVER.....16	@.....9
c!.....9	pile de retour.....8	*.....14
c@.....9	R>.....17	*/.....15
constant.....10	ROT.....17	*/MOD.....15
drop.....8	SWAP.....16	/.....14
dup.....8	U.....13	/MOD.....15
DUP.....16	value.....10	+.....13
forget.....6	variable.....9	>R.....17
import.....19	web.....19	
JSWORD:.....216	